

智能垃圾分类处理系统仿真程序设计

成绩

指导教师：罗志祥

专业班级：光电信息科学与工程 202409

姓名：陈恪瑾

学号：U202413925

联系电话：18271851291

1 任务要求

1.1 实验目的

- 掌握使用 AT89S52 的 I/O 口；
- 理解 AT89S52 单片机的基本结构和工作原理；
- 掌握分支结构，条件转移程序设计方法；
- 具备调试程序的基本技能。

1.2 实验内容

在 Keil 环境中设计一个基于 AT89S52 单片机的智能垃圾分类处理系统，用户通过 P2 口的输入和 P1 口输出（上电复位初始状态为全 1），模拟小区垃圾回收终端系统的运行状态，系统需要实现以下功能：

- 小区居民通过键入相应按键选择垃圾类型（分别用 P2.4 P2.7 端口模拟按键输入，低电平有效，P2.7=0、P2.6=0、P2.5=0、P2.4=0 分别代表“塑料纸张”、“玻璃金属”、“厨余垃圾”和“危害垃圾”四类）。
- P1.0 P1.3 四位代表垃圾桶盖开关控制，控制名为“塑料纸张”、“玻璃金属”、“厨余垃圾”和“危化品垃圾”四个垃圾桶盖，高电平为关闭状态，低电平为打开状态。
- 垃圾桶盖每次打开的时间缺省值为 8s，然后自动关闭，请采用循环程序配合 NOP 指令完成延时子程序，精度控制在 0.1s 以内。

请编写程序，实现用户键入，控制相应的垃圾桶盖打开和关闭。

1.3 提高部分（申优选做）

- 系统初始化后，随机生成 4 个 0 9 的整数，置于寄存器 R0 R3 中表示四个垃圾桶的剩余容量。用户每次投递之后，容量相应减少。当投放垃圾超出剩余容量后，对

应的标志位发生变化（自行定义），便于提醒物业及时清理，此时再次按键投放时，该桶盖不再打开避免溢出。

- 其他个性化设计。

二、设计思路

本系统基于 AT89S52 单片机核心架构，采用模块化程序设计思想，围绕垃圾分类投放的核心需求，拆分功能模块实现全流程控制，整体设计逻辑如下：

1. **系统初始化模块：**上电后完成核心状态配置，包括 P1 口置全 1（所有桶盖默认关闭）、P2 口置全 1（配置为输入模式）；调用伪随机数生成子程序，为 4 个垃圾桶生成 0-9 的初始剩余容量；清零所有溢出告警标志位，完成系统初始状态复位。
2. **按键检测与分支跳转模块：**主循环中持续读取 P2 口状态，通过取反、屏蔽低 4 位操作判断是否有按键按下；无按键时持续循环检测，有按键时通过位判断指令，跳转到对应垃圾类型的处理分支，实现多分支条件转移控制。
3. **垃圾投放与容量管理模块：**按键触发后，先校验对应垃圾桶的剩余容量：若容量为 0，直接置位对应溢出标志位，返回主循环，禁止开盖；若容量非 0，拉低对应 P1 口引脚打开桶盖，调用 8s 延时子程序后拉高引脚关闭桶盖，同时将对应容量寄存器递减 1，完成一次完整投放流程。
4. **精准延时子程序：**基于 12MHz 晶振（1 个机器周期 = $1\mu s$ ），采用三层嵌套循环配合 NOP 指令，精准计算每个指令的机器周期，实现 8s 精准延时，精度控制在 0.1s 以内，满足桶盖自动关闭的时序要求。
5. **伪随机数生成子程序：**采用互质步长累加取余算法，以片内 RAM 30H 单元为随机数种子，选用与 10 互质的 7 作为固定步长，通过累加后对 10 取余的方式，生成 4 个 0-9 范围内的伪随机数，实现垃圾桶初始容量的随机初始化。

三、资源分配

(一) I/O 口硬件资源分配

I/O 引脚	方向	功能定义	有效电平
P1.0	输出	塑料纸张垃圾桶盖控制	低电平打开，高电平关闭
P1.1	输出	玻璃金属垃圾桶盖控制	低电平打开，高电平关闭
P1.2	输出	厨余垃圾垃圾桶盖控制	低电平打开，高电平关闭
P1.3	输出	危化品垃圾垃圾桶盖控制	低电平打开，高电平关闭
P2.4	输入	危害垃圾投放按键	低电平有效
P2.5	输入	厨余垃圾投放按键	低电平有效
P2.6	输入	玻璃金属投放按键	低电平有效
P2.7	输入	塑料纸张投放按键	低电平有效

(二) 寄存器资源分配

寄存器	功能定义
R0	存储塑料纸张垃圾桶剩余容量，初始值为 0-9 随机数
R1	存储玻璃金属垃圾桶剩余容量，初始值为 0-9 随机数
R2	存储厨余垃圾垃圾桶剩余容量，初始值为 0-9 随机数
R3	存储危化品垃圾垃圾桶剩余容量，初始值为 0-9 随机数
R5-R7	8s 延时子程序三层循环计数寄存器，用于精准控制延时时长
累加器 A	临时存储 I/O 口状态、随机数运算结果、数据中转

(三) 片内 RAM 资源分配

存储单元	功能定义
30H 单元	伪随机数生成子程序的种子存储单元，用于连续生成 0-9 随机数
20H.0	塑料纸张垃圾桶溢出告警标志位，置 1 表示容量用尽
20H.1	玻璃金属垃圾桶溢出告警标志位，置 1 表示容量用尽
20H.2	厨余垃圾垃圾桶溢出告警标志位，置 1 表示容量用尽
20H.3	危化品垃圾垃圾桶溢出告警标志位，置 1 表示容量用尽

四、流程图

(请用 Visio/Draw.io 绘制后，插入\includegraphics命令)

五、源代码

```

; =====
; 文件名称: SmartTrash.asm
; 处理器: AT89S52 (12MHz晶振)
; 实验内容: 智能垃圾分类处理系统仿真程序
; =====

    ORG 0000H

    LJMP MAIN          ; 上电跳转至主程序, 避开中断向量区

    ORG 0030H          ; 主程序起始地址, 避开中断入口
; =====
; 主程序: 系统初始化+主循环按键检测
; =====

MAIN:

    ; 1. 系统I/O口初始化

    MOV P1, #0FFH      ; P1.0~P1.3 上电为全1, 所有垃圾桶盖默认关闭
    MOV P2, #0FFH      ; P2口置全1, 配置为输入模式, 准备检测按键

    ; 2. 生成4个0~9伪随机数, 作为四个垃圾桶初始剩余容量

    MOV 30H, #03H      ; 给30H单元设置初始种子, 用于伪随机数生成
    LCALL GET_RAND_0_9

    MOV R0, A           ; R0 = 塑料纸张垃圾桶初始容量 (0-9随机数)
    LCALL GET_RAND_0_9

    MOV R1, A           ; R1 = 玻璃金属垃圾桶初始容量 (0-9随机数)
    LCALL GET_RAND_0_9

    MOV R2, A           ; R2 = 厨余垃圾垃圾桶初始容量 (0-9随机数)
    LCALL GET_RAND_0_9

    MOV R3, A           ; R3 = 危化品垃圾垃圾桶初始容量 (0-9随机数)

    ; 3. 溢出告警标志位初始化, 全部清零

    CLR 20H.0           ; 20H.0 = 塑料纸张桶溢出标志
    CLR 20H.1           ; 20H.1 = 玻璃金属桶溢出标志

```

CLR 20H.2 ; 20H.2 = 厨余垃圾桶溢出标志

CLR 20H.3 ; 20H.3 = 危化品桶溢出标志

; 主循环：持续检测按键输入

MAIN_LOOP:

MOV A, P2 ; 读取P2口完整状态

CPL A ; 按键低电平有效，取反后按下的按键对应位变为1

ANL A, #0F0H ; 屏蔽低4位，仅保留P2.4~P2.7按键相关位

JZ MAIN_LOOP ; 无按键按下（全0），返回循环继续检测

; 多分支查询，判断具体按下的按键，跳转对应处理分支

JB ACC.7, THROW_PLASTIC ; ACC.7对应P2.7，塑料纸张按键按下

JB ACC.6, THROW_GLASS ; ACC.6对应P2.6，玻璃金属按键按下

JB ACC.5, THROW_KITCHEN ; ACC.5对应P2.5，厨余垃圾按键按下

JB ACC.4, THROW_HAZARD ; ACC.4对应P2.4，危化品垃圾按键按下

SJMP MAIN_LOOP ; 无有效按键，返回主循环

; =====

; 垃圾投放处理分支模块

; =====

; 塑料纸张垃圾投放处理

THROW_PLASTIC:

CJNE R0, #00H, OPEN_PLAS ; 剩余容量不为0，跳转开盖流程

SETB 20H.0 ; 容量为0，置位溢出标志，提醒物业清理

SJMP MAIN_LOOP ; 禁止开盖，返回主循环

OPEN_PLAS:

CLR P1.0 ; 拉低P1.0，打开塑料纸张桶盖

LCALL DELAY_8S ; 调用8s延时子程序，保持桶盖打开

SETB P1.0 ; 拉高P1.0，关闭桶盖

DEC R0 ; 投递完成，剩余容量减1

SJMP MAIN_LOOP ; 返回主循环，等待下一次按键

；玻璃金属垃圾投放处理

THROW_GLASS:

```
CJNE R1, #00H, OPEN_GLAS ; 剩余容量不为0, 跳转开盖流程
SETB 20H.1                ; 容量为0, 置位溢出标志
SJMP MAIN_LOOP            ; 禁止开盖, 返回主循环
```

OPEN_GLAS:

```
CLR P1.1                  ; 拉低P1.1, 打开玻璃金属桶盖
LCALL DELAY_8S            ; 8s延时
SETB P1.1                 ; 关闭桶盖
DEC R1                    ; 剩余容量减1
SJMP MAIN_LOOP
```

；厨余垃圾投放处理

THROW_KITCHEN:

```
CJNE R2, #00H, OPEN_KITC ; 剩余容量不为0, 跳转开盖流程
SETB 20H.2                ; 容量为0, 置位溢出标志
SJMP MAIN_LOOP            ; 禁止开盖, 返回主循环
```

OPEN_KITC:

```
CLR P1.2                  ; 拉低P1.2, 打开厨余垃圾桶盖
LCALL DELAY_8S            ; 8s延时
SETB P1.2                 ; 关闭桶盖
DEC R2                    ; 剩余容量减1
SJMP MAIN_LOOP
```

；危化品垃圾投放处理

THROW_HAZARD:

```
CJNE R3, #00H, OPEN_HAZA ; 剩余容量不为0, 跳转开盖流程
SETB 20H.3                ; 容量为0, 置位溢出标志
SJMP MAIN_LOOP            ; 禁止开盖, 返回主循环
```

OPEN_HAZA:

```
CLR P1.3                  ; 拉低P1.3, 打开危化品桶盖
LCALL DELAY_8S            ; 8s延时
```

```

        SETB P1.3                ; 关闭桶盖
        DEC R3                    ; 剩余容量减1
        SJMP MAIN_LOOP

; =====
; 子程序名: DELAY_8S
; 功 能: 实现8s精准延时, 精度控制在0.1s以内
; 参 数: 无
; 原 理: 12MHz晶振下, 1个机器周期=1us
;         内循环: 250次 × 4us = 1ms
;         中层循环: 200次 × 1ms = 200ms
;         外层循环: 40次 × 200ms = 8000ms = 8s
; =====
DELAY_8S:
        MOV R5, #40              ; 最外层循环, 40次 × 200ms = 8s
D_OUTER:
        MOV R6, #200             ; 中层循环, 200次 × 1ms = 200ms
D_MIDDLE:
        MOV R7, #250             ; 内层循环, 250次 × 4us = 1ms
D_INNER:
        NOP                      ; 1个机器周期, 1us
        NOP                      ; 1个机器周期, 1us
        DJNZ R7, D_INNER         ; 2个机器周期, 2us; 单次内循环共4us
        DJNZ R6, D_MIDDLE        ; 中层循环跳转
        DJNZ R5, D_OUTER         ; 外层循环跳转
        RET                      ; 延时完成, 子程序返回

; =====
; 子程序名: GET_RAND_0_9
; 功 能: 生成0-9范围内的伪随机数
; 参 数: 无, 返回值存于累加器A
; 原 理: 选用与10互质的7作为步长, 累加后对10取余,

```

```

;          保证生成0-9完整循环的伪随机数，无重复死区
; =====
GET RAND_0_9:
    MOV A, 30H          ; 读取当前随机数种子
    ADD A, #07H         ; 累加固定步长7

MOD_10:
    CLR C              ; 清除借位标志
    SUBB A, #0AH       ; 减去10，实现取余运算
    JNC MOD_10         ; 无借位 (A>=10)，继续减10
    ADD A, #0AH        ; 有借位，补回多减的10，得到0-9余数
    MOV 30H, A         ; 更新随机数种子，用于下次生成
    RET               ; 随机数存于A，子程序返回

    END               ; 程序结束

```

六、程序测试方法与结果

采用 keil 软件进行仿真测试。

效果如下：

将 P2.6 调为低电平时，观察到 P1.1 也变为低电平，表示打开玻璃金属垃圾桶。

(此处插入对应图片：\includegraphics[width=10cm]{图片路径})

下图为延时，由光标可以看到大致为 8s。

(此处插入对应图片：\includegraphics[width=10cm]{图片路径})

下图可以看到，当 R0 为 0 时（表示第一个垃圾桶已装满），此时调节 P2.7=0，并不会使 P1.0 打开。

(此处插入对应图片：\includegraphics[width=10cm]{图片路径})

七、实验问题分析与对策

(根据实验实际情况填写)

本人承诺: 本报告内容真实，无伪造数据，无抄袭他人成果。本人完全了解学校相关规

定，如若违反，愿意承担其后果。

签字：

陈恪瑾

2026 年 4 月 26 日